We start with a model of co-repression taken from Buchler *et al*. In particular, we allow for a degree of repressor interaction captured by ω. There is no explicit term for leaky expression, since repression was found to completely shut off the strongest promoters. Furthermore, we assume that the induction is perfect, so the effect of a repressor can be removed completely.

```
P[R1_, R2_, c1_, c2_, ω_] :=
                      A
  ─────────────────────────────────────────
  1 + c1 * R1 + c2 * R2 + ω * c1 * c2 * R1 * R2
r[c1_, c2_, ω_] :=
 Log[P[0, 0, c1, c2, ω] / P[1, 1, c1, c2, ω]]
α[c1_, c2_, ω_] :=
 Log[P[0, 0, c1, c2, ω] / P[1, 0, c1, c2, ω]] /
  r[c1, c2, ω]
β[c1_, c2_, ω_] :=
 Log[P[0, 0, c1, c2, ω] / P[0, 1, c1, c2, ω]] /
  r[c1, c2, ω]
```

```
l[c1_, c2_, ω_] = (α[c1, c2, ω] + β[c1, c2, ω]) / 2
```

$$\frac{1}{2}\left(\frac{\mathrm{Log}[1+c1]}{\mathrm{Log}[1+c1+c2+c1\,c2\,\omega]}+\frac{\mathrm{Log}[1+c2]}{\mathrm{Log}[1+c1+c2+c1\,c2\,\omega]}\right)$$

Note that if ω=1 then l=0.5. This means than independent repressor binding can only generate SIG, SLOPE or asym-SLOPE logic. This could result in apparent AND logic if the repression by each was so strong as to lower it below background.

```
Simplify[l[c1, c2, 1], {c1 > 0, c2 > 0}]
```

$$\frac{1}{2}$$

```
a[c1_, c2_, ω_] = α[c1, c2, ω] - β[c1, c2, ω]
```

$$\frac{\text{Log}[1+\text{c1}]}{\text{Log}[1+\text{c1}+\text{c2}+\text{c1 c2}\,\omega]}-\frac{\text{Log}[1+\text{c2}]}{\text{Log}[1+\text{c1}+\text{c2}+\text{c1 c2}\,\omega]}$$

Note that clearly for a=0 then c1=c2. Therefore two balanced independently interacting repressors will always generate a slope gate. If we now examine this case of balanced repression.

```
l[c, c, ω]
```

$$\frac{\text{Log}[1+c]}{\text{Log}[1+2\,c+c^2\,\omega]}$$

```
a[c, c, ω]
```

0

```
r[c, c, ω]
```

$$\text{Log}[1+2\,c+c^2\,\omega]$$

First suppose that the two repressors are exclusive ω=0.

```
Plot[l[c, c, 0], {c, 0, 1000}];
```



Now examine the relationship between repression and logic as a function of the binding constant c and the interaction ω.

$$\text{solutions} = \text{Solve}[\text{Log}[1 + 2\,c + c^2\,\omega] == rs,\ \omega]$$

$$\left\{\left\{\omega \to \frac{-1 - 2\,c + e^{rs}}{c^2}\right\}\right\}$$

$$l[c,\ c,\ \omega]\ /.\ \text{solutions}$$

$$\left\{\frac{\text{Log}[1 + c]}{\text{Log}[e^{rs}]}\right\}$$

$$\text{solutions} = \text{Solve}\left[\frac{\text{Log}[1 + c]}{rs} == ls,\ c\right]$$

$$\{\{c \to -1 + e^{ls\,rs}\}\}$$

```
(-1 - 2 c + ⅇ^rs)/c^2 /. solutions
```

$$\left\{\frac{-1+e^{rs}-2\,(-1+e^{ls\,rs})}{(-1+e^{ls\,rs})^{2}}\right\}$$

What does ω look like as a function of r and l?

```
Plot3D[(-1 + ⅇ^rs - 2 (-1 + ⅇ^(ls rs)))/(-1 + ⅇ^(ls rs))^2, {rs, 1, 10},
  {ls, .1, 1}, AxesLabel → {r, l, ω}, ColorFunction → Hue,
  Mesh → False, ViewPoint → {-18, -2, 5}];
```




```
25
```

Note that for a perfect OR gate (r >>1 and l=0), we would have to have $\omega\to\infty$. Furthermore, to get even a decent dual-repressor OR (R > 10, l < 0.1) we need to have an extremely strong cooperative interaction $\omega\approx100$. In the case of balanced repression it is easy to calculate the repression constant and interaction term given the logic type and regulatory range. Now consider the asymmetric logic gates where we fix the regulatory range. Then we can solve for ω in terms of the two repression constants and use this to determine (a,l) as a function of these constants. Therefore, if we know the repression constants (from analysis of SIGs), then the interaction term will determine what types of logic are possible.

```
solutions = Solve[r[c1, c2, ω] == r, ω]
```

$$\left\{\left\{\omega \to \frac{-1 - c1 - c2 + e^r}{c1\,c2}\right\}\right\}$$

$$\text{Simplify}\left[l\left[c1, c2, \frac{-1 - c1 - c2 + e^r}{c1\,c2}\right], r > 0\right]$$

$$\frac{\text{Log}[1 + c1] + \text{Log}[1 + c2]}{2\,r}$$

$$\text{Simplify}\left[a\left[c1, c2, \frac{-1 - c1 - c2 + e^r}{c1\,c2}\right], r > 0\right]$$

$$\frac{\text{Log}[1 + c1] - \text{Log}[1 + c2]}{r}$$